

Tanger と Pitzer の式を用いた塩化ナトリウム水溶液の飽和水蒸気圧を求める計算プログラム

300°Cを超える高温条件での塩化ナトリウム水溶液の飽和水蒸気圧に関するTangerとPitzerの式を本サイト(<http://www.hyogo-u.ac.jp/sci/yshibue/solution.html>)内の文書で解説した。飽和水蒸気圧の計算プログラムをテキスト形式で保存して本サイト内で示している。このプログラムは澁江(2012, p. 61–63)が示したものに数箇所の修正を加えたものである。

ここでは、計算プログラムへの入力と出力、プログラム中の変数について解説する。そして、最後にプログラムのリストを示す。文書の1行文字数の関係で、プログラムの1つのlineが文書では複数行にまたがっていることがある。プログラムでは行番号が先頭に付くので、行番号ごとにひとまとまりのlineになっていると理解してほしい。

1. 入力と出力

この文書ではプログラムへの入力例と出力例を示す。そして、プログラム中で用いている変数を説明する。プログラムを起動すると、以下の問いが出てくる。問いを(1), (2), (3), (4)で示し、入力方法をそれぞれについて説明する。

(1) “Input temperature (°C)?”

摂氏温度で温度を入力する。入力した温度が250°C未満だと、プログラムは温度の再入力を求めるようになっている。

(2) “Input molality of NaCl in liquid phase?”

液相中での塩化ナトリウムの質量モル濃度を入力する。

(3) “Do you input initial estimates for DRL, DRV, and YV? Input Y (or y) or N (or n)?”

この問いは、液相中での水の密度と関連する値である r^{liquid} 値(DRL), 気相中での水の密度と関連する値である r^{vapor} 値(DRV), 気相中での塩化ナトリウムの y 値(YV)を全て入力するかどうかに関するものである。これら3つの未知数についての初期推定値を手入力する時はYあるいはyを入力する。これ以外の場合はNあるいはnを入力する。

手入力すると答えた場合には、まず、“DRL?”と問いかけがあるので r^{liquid} 値を入力する。次に、“DRV?”と問いかけがあるので r^{vapor} 値を入力する。次に、“YV?”と問いかけがあるので y^{vapor} 値を入力する。Nあるいはnと入力すると、初期推定値をプログラムが計算する。

(4) 全ての計算が終了すると、新たに“Input temperature (°C)?”と問いが出てくる。続けて計算する時は(1)から(3)の要領で入力していく。計算を終了させる時は、“Ctrl”と“c”を同時に押す。

温度を350°C, 濃度を1 mol kg⁻¹, 初期推定値の手入力を行わないとした時の入力と出力をそれぞれ表1と表2に示す。

表1 温度を350°C, 濃度を1 mol kg⁻¹, 初期推定値の手入力を行わないとした時の入力画面

```
run
Input temperature (deg C)? 350
Input molality of NaCl in liquid phase? 1
Do you input initial estimates for DRL, DRV, and YV? Input Y (or y) or N (or n)? n

Input temperature (deg C)?
```

表2 温度を350°C, 濃度を1 mol kg⁻¹, 初期推定値の手入力を行わないとした時の出力画面

```

Temperature(° C)=350.000      molality of NaCl= +1.000000

Eps=+1.00D-005
DRLMAX=+3.50 DRLMIN=+1.50 DRVMAX=+1.00 DRVMIN=+0.02 YVMAX=+1.70D-002 YVMIN=+1.00D-012
Initial estimate
BC(1)=1.9816D+000 BC(2)=+3.5267D-001 BC(3)=+3.1187D-005
Initial FSUM=+3.272784D-004

ICON= 3 FSUM=+3.955443D-005 ITERATION= 1
BC(1)=1.9777D+000 BC(2)=+3.2691D-001 BC(3)=+1.9561D-005

ICON= 3 FSUM=+6.207883D-006 ITERATION= 2
BC(1)=1.9741D+000 BC(2)=+3.1140D-001 BC(3)=+1.5930D-005

ICON= 3 FSUM=+5.674276D-009 ITERATION= 3
BC(1)=1.9742D+000 BC(2)=+3.1265D-001 BC(3)=+1.6430D-005

ICON= 2 FSUM=+7.616774D-016 ITERATION= 4
BC(1)=1.9742D+000 BC(2)=+3.1266D-001 BC(3)=+1.6439D-005

ICON= 0 FSUM=+5.348936D-025 ITERATION= 5
BC(1)=1.9742D+000 BC(2)=+3.1266D-001 BC(3)=+1.6439D-005

Solution
BC(1)= +1.9741581841D+000
BC(2)= +3.1265655698D-001
BC(3)= +1.6439159829D-005

Pressure liquid=+1.590D+002 vapor=+1.590D+002
Density liquid=+6.728D-001 vapor=+1.007D-001

```

表2で示した出力について説明を加える。

- (1) 1行目の温度と濃度は入力値を示す。
- (2) 3行目は、逐次近似計算の収束条件 (Epsの値) を示す。値は10のべき乗を用いる指数形式で表している。1.00D-005としているのでEpsは10⁻⁵である。以下に示すD+00n (nは任意の値) は10ⁿ, D-00nは10⁻ⁿを表している。
- (3) 4行目は、逐次近似計算の際のDRLとDRVとYVの最大値と最小値を示している。DRLの最大値をDRLMAX, DRLの最小値をDRLMIN, DRVの最大値をDRVMAX, DRVの最小値をDRVMIN, YVの最大値をYVMAX, YVの最小値をYVMINとして表している。YVMAXの値は, $y^{\text{liquid}} - 0.001$ に取り, YVMINの値を10⁻¹²に取っている。YVMAXの値を指定しないと, 収束値が y^{liquid} と等しくなる場合が出てくるのでこのようにした。その他の値は温度によって変えている。350°C以下では, DRLMAX=3.5, DRLMIN=1.5, DRVMAX=1, DRVMIN=0.02にしている。350°Cより高温で540°C以下では, DRLMIN=0.5, DRVMAX=2, 540°Cより高温では, DRLMIN=0.5, DRVMAX=2.5にしている。
- (5) 5行目と6行目は, DRL, DRV, YVに関する初期推定値を示している。BC(1)はDRL, BC(2)はDRV, BC(3)はYVの初期推定値である。
- (6) 7行目は, 初期推定値を用いて求めた次の計算値(FSUM)を示している。

$$FSUM = \left(\frac{p^{\text{vapor}}}{p^{\text{liquid}}} - 1 \right)^2 + \left(\frac{\mu_w^{\text{vapor}}}{\mu_w^{\text{liquid}}} - 1 \right)^2 + \left(\frac{\mu_s^{\text{vapor}}}{\mu_s^{\text{liquid}}} - 1 \right)^2$$

- (7) 9行目と10行目は, 初期推定値から求められる新しい推定値とFSUMの値を示している。ICONとして示した値は収束条件を満たしていない未知数の数を示し, ITERATIONとして示した数は逐次近似計

算の回数を表す。12行目から22行目についても同様である。

(8) 24行目から27行目は、DRL, DRV, YVの計算結果を示している。BC(1)は r^{liquid} , BC(2)は r^{vapor} , BC(3)は y^{vapor} の計算結果である。

(9) 29行目は、DRL, DRV, YVの計算結果を用いて計算した圧力(単位はbar)を液相と気相に分けて示している。

(10) 30行目は、DRLとDRVの計算結果を用いて計算した液相と気相の密度(単位は g cm^{-3})を液相と気相に分けて示している。なお、逐次近似の回数が100回を超えてしまった場合には、BC(1), BC(2), BC(3)の計算結果に「Function improvement possible」と印字するようになっている。

入力段階で過飽和状態になる濃度を入力した場合には、DRLとDRVとYVに関する問いかけの前に計算が始まる。例えば、350°Cで塩化ナトリウムの質量モル濃度を20と入力した場合、計算が始まって表3のように出力が出る。

表3 温度を350°C, 濃度を20 mol kg⁻¹とした時の出力画面

Temperature(° C)=350.000 molality of NaCl=+20.000000

Supersaturated (or saturated) with NaCl

Pressure = +1.064+002 Solubility=+12.37

表3で示した出力について説明を加える。

(1) 1行目の温度と濃度は入力値を示す。

(2) 3行目は、塩化ナトリウムに対して過飽和状態あるいは飽和状態であることを表している。

(3) 4行目は、Bischoff (1991)から計算した飽和水蒸気圧が106.4 barであることと、この温度における飽和濃度が12.37 mol kg⁻¹であることを表している。

2. プログラム使用上の留意点

この計算プログラムではHaar et al. (1984)の状態方程式を用いて計算した圧力の単位をbarに取っており、MPaからの換算のために変数FPを用いている。10 cm³ bar = 1 Jの関係を用いて、体積と圧力の積をエネルギーに換算するために、line 10450で変数VC(純水の臨界モル体積)の値を10で割っている。

気液平衡計算を行う時に、圧力や液相中と気相中の水と塩化ナトリウムの化学ポテンシャルの計算値を液相と気相の密度や気相に含まれている塩化ナトリウムの物質量(モル)で偏微分する計算が必要となる。本計算プログラムでは、これらすべての偏微分係数の計算値を差分で求めている。

気相における塩化ナトリウムのy値の上限(YVMAX)をYVMAX = YL - 0.001で与えている(リスト中のline 22600)。したがって、気液二相平衡状態の計算も $y^{\text{liquid}} - y^{\text{vapor}} \geq 0.001$ の領域に限定されている。つまり、臨界点の計算や臨界点付近での計算を行うことができない。line 22600で示した計算式を変えることで臨界点付近における気液二相平衡状態の計算が可能になる。ただし、YVMAXがYLと等しいとおくと、臨界点ではないのに気相の組成が液相の組成と等しくなって計算が終了することがある。本計算プログラムで使用する気体定数の値は、Haar et al. (1984)が用いた値とは異なっている。そこで、質量モル濃度として0を入力した時(あるいは誤って負の値を入力した時)には質量モル濃度を再入力するようにしている。

解説で記したように、液相の密度の初期推定値を求める計算式は350°C程度まででの実験値から求めている。したがって、初期推定値を求める計算式を350°Cを超える温度で使用すると、初期推定値の不確かさが大きくなり、計算に失敗することがある。このような場合には、濃度はそのままにして温度を350°Cあるいはそれ以下に指定して、いったん低温での飽和水蒸気圧を計算する。求めることができたDRL, DRV, YVの値を用いて、温度だけを少し高くして計算する。新たに求めることができたDRL, DRV, YVの値を用いてさらに温度を高くする。以上の操作を繰り返して、求めようとしている温度での飽和水蒸気圧を計算する。臨界点付近では、1°C刻みで温度を上げていくようなこと

をしないと，計算が収束しない。

3. プログラム中の変数

プログラムで用いている変数（文字変数とFORループの制御変数を除く）の意味を表にして以下に示す。変数の中にはプログラム中で二通りの意味で用いられているものがある。そこで、このような変数については、変数名の後の括弧内に行番号を示している。なお、配列変数の添字に大文字のIを用いているが、その添字は係数に付けている下付き文字*i*と同義である。また、HGK式で用いられている変数やHGK式を用いて計算している変数は「変数の意味」欄に「(HGK式)」と記している。

プログラム中の変数	変数の意味
AA	1 (HGK式)
AAD(I)	α_i (HGK式)
AAT(I)	β_i (HGK式)
AB	A_{base}/RT (HGK式)
AD	A/RT (HGK式)
ADZ(I)	ρ_i (HGK式)
AI	$A_{\text{ideal gas}}/RT$ (HGK式)
AL	解の計算に取り入れる縮小因子
AR	A_{residual}/RT (HGK式)
ATT	β_i (HGK式)
ATZ(I)	T_i (HGK式)
B1	b (HGK式)
B10	Tanger and Pitzer (1989)の状態方程式中の係数
B11	Tanger and Pitzer (1989)の状態方程式中の係数
B2	\bar{B} (HGK式)
B20	Tanger and Pitzer (1989)の状態方程式中の係数
BA1,BA2,BA3,BA4, BA5	塩化ナトリウム水溶液の密度に関するHaas式で使用する係数
BASEF	$p_{\text{base}}/\rho RT$ (HGK式)
BBC(1)	r^{liquid} の計算値
BBC(2)	r^{vapor} の計算値
BBC(3)	y^{vapor} の計算値
BC(1)	r^{liquid} の計算値
BC(2)	r^{vapor} の計算値
BC(3)	y^{vapor} の計算値
BP(1)	b_0 (HGK式)
BP(2)	b_1 (HGK式)
BP(3)	b_2 (HGK式)
BP(4)	0 (HGK式)
BP(5)	b_3 (HGK式)

BP(6)	b_4 (HGK式)
BP(7)	b_5 (HGK式)
BP(8)	0 (HGK式)
BP(9)	0 (HGK式)
BP(10)	0 (HGK式)
BQ(1)	B_0 (HGK式)
BQ(2)	0 (HGK式)
BQ(3)	B_1 (HGK式)
BQ(4)	B_2 (HGK式)
BQ(5)	B_3 (HGK式)
BQ(6)	B_4 (HGK式)
BQ(7)	B_5 (HGK式)
BQ(8)	0 (HGK式)
BQ(9)	0 (HGK式)
BQ(10)	0 (HGK式)
C(I)	C_i (HGK式)
COEF(I,J)	最小二乗法による計算に用いる行列
D	計算過程で用いる密度の値 (HGK式)
D1	Haas式から求められる水1 kgを含む水溶液の体積(cm ³)
DAQ	Haas式から求められる塩化ナトリウム水溶液の密度(g cm ⁻³)
DD	密度の計算値 (HGK式)
DDZ	ρ_i (HGK式)
DEL	δ_i (HGK式)
DELDL	数値微分するための変数(10 ⁻⁶)
DELDV	数値微分するための変数(10 ⁻⁶)
DEX	$\delta_i^j \exp(-\alpha_i \delta_i^{k_i})$ (HGK式)
DIFMYUS	$\mu_s^{\text{vapor}} - \mu_s^{\text{liquid}}$
DIFMYUWATER	$\mu_w^{\text{vapor}} - \mu_w^{\text{liquid}}$
DIFP	$p^{\text{vapor}} - p^{\text{liquid}}$
DIFPLV	$p^{\text{liquid}} - p^{\text{vapor}}$
DL	計算過程で用いる液相の密度の値
DMYUSDDL	$\left(\frac{\partial \mu_s^{\text{liquid}}}{\partial r^{\text{liquid}}} \right)_{T, r^{\text{vapor}}, y^{\text{vapor}}}$
DMYUSDDV	$\left(\frac{\partial \mu_s^{\text{vapor}}}{\partial r^{\text{vapor}}} \right)_{T, r^{\text{liquid}}, y^{\text{vapor}}}$

DMYUSDY	$\left(\frac{\partial \mu_s^{\text{vapor}}}{\partial y^{\text{vapor}}}\right)_{T, r^{\text{liquid}}, r^{\text{vapor}}}$
DMYUWATERDDL	$\left(\frac{\partial \mu_w^{\text{liquid}}}{\partial r^{\text{liquid}}}\right)_{T, r^{\text{vapor}}, y^{\text{vapor}}}$
DMYUWATERDDV	$\left(\frac{\partial \mu_w^{\text{vapor}}}{\partial r^{\text{vapor}}}\right)_{T, r^{\text{liquid}}, y^{\text{vapor}}}$
DMYUWATERDY	$\left(\frac{\partial \mu_w^{\text{vapor}}}{\partial y^{\text{vapor}}}\right)_{T, r^{\text{liquid}}, r^{\text{vapor}}}$
DPDDL	$\left(\frac{\partial p^{\text{liquid}}}{\partial \rho^{\text{liquid}}}\right)_{T, y^{\text{liquid}}}$
DPDDV	$\left(\frac{\partial p^{\text{vapor}}}{\partial \rho^{\text{vapor}}}\right)_{T, y^{\text{vapor}}}$
DPDY	$\left(\frac{\partial p^{\text{vapor}}}{\partial y^{\text{vapor}}}\right)_{T, r^{\text{vapor}}}$
DRL	液相の r 値
DRL0	Haas式から求められる液相の r 値
DRLDEC	DRLを求めるための変数
DRLMAX	DRLの最大値
DRLMIN	DRLの最小値
DRV	気相の r 値
DRV0	Wagner and Pruss (1993)の式で近似した気相の r 値
DRVMAX	DRVの最大値
DRVMIN	DRVの最小値
DV	計算過程で用いる気相の密度の値
DWATERV	Wagner and Pruss (1993)の式で計算した純水だけからなる気相の密度
E	$e^{-\rho}$ (HGK式)
EPS	逐次近似計算の収束条件
EX1	$-\alpha_i \delta_i^{k_i}$ (HGK式)
EX2	$-\beta_i \tau_i^2$ (HGK式)
FCT	$\frac{\delta_i^l}{\rho_i} \left[l_i \delta_i^{-1} - \alpha_i k_i \delta_i^{k_i-1} \right] \rho^2 \exp(-\alpha_i \delta_i^{k_i} - \beta_i \tau_i^2)$ (HGK式)
FD	1.0 (HGK式)
FH	18.01534
FP	圧力の単位をbarにするための換算定数(10.0) (HGK式)

FSUM	$\left(\frac{p^{\text{vapor}}}{p^{\text{liquid}}}-1\right)^2 + \left(\frac{\mu_w^{\text{vapor}}}{\mu_w^{\text{liquid}}}-1\right)^2 + \left(\frac{\mu_s^{\text{vapor}}}{\mu_s^{\text{liquid}}}-1\right)^2$
FSUMI	逐次近似計算の過程における $\left(\frac{p^{\text{vapor}}}{p^{\text{liquid}}}-1\right)^2 + \left(\frac{\mu_w^{\text{vapor}}}{\mu_w^{\text{liquid}}}-1\right)^2 + \left(\frac{\mu_s^{\text{vapor}}}{\mu_s^{\text{liquid}}}-1\right)^2$
G1	11 (HGK式)
G2	133/3 (HGK式)
GASCON	気体定数を水のモル質量で割った値(0.461522 J g ⁻¹ K ⁻¹)。
GD	G/RT (HGK式)
GF	7/2 (HGK式)
GI	A _{ideal gas} を計算するための式で用いる変数 (HGK式)
GIB	純水のギブスエネルギー
GIBL	純水が気液二相平衡条件下にある時の液相のギブスエネルギー
GIBLD	密度がRHOLDの時の純水のギブスエネルギー
GIBV	純水が気液二相平衡条件下にある時の気相のギブスエネルギー
GIBVD	密度がRHOVDの時の純水のギブスエネルギー
H2O	水のモル質量(= 18.01534 g mol ⁻¹)
HGKG(I)	g _i (HGK式)
ICON	収束条件を満たしていない未知数の数
II(I)	k _i - 1 (i = 1, ..., 36)あるいはk _i (i = 37, ..., 40) (HGK式)
IMAX	逐次近似計算の上限回数
INC	36 (HGK式)
ITER	逐次近似計算の回数
JJ(I)	l _i + 1 (i = 1, ..., 36)あるいはl _i (i = 37, ..., 40) (HGK式)
K	k _i (HGK式)
KK	未知数の数
KM	l _i (HGK式)
KP1	未知数の数に1を加えた値
L	l _i + 1 (HGK式)
M	質量モル濃度
MYULS	液相中での塩化ナトリウムの化学ポテンシャルの値からg*を引いたもの
MYUVS	気相中での塩化ナトリウムの化学ポテンシャルの値からg*を引いたもの
MYULWATER	液相中での水の化学ポテンシャル(J mol ⁻¹)
MYUVWATER	気相中での水の化学ポテンシャル(J mol ⁻¹)
NACL	塩化ナトリウムのモル質量(= 58.4428 g mol ⁻¹)
PBISCHOFF	気相と液相と固相が平衡状態にある時の飽和水蒸気圧

PL	液相の組成と密度の推定値から計算できる飽和水蒸気圧
PLD	$\left(\frac{\partial p^{\text{liquid}}}{\partial r^{\text{liquid}}}\right)_{T,y^{\text{liquid}}}$ を求めるための変数
PRES	p (HGK式)
PV	気相の組成と密度の推定値から計算できる飽和水蒸気圧
PVD	$\left(\frac{\partial p^{\text{vapor}}}{\partial r^{\text{vapor}}}\right)_{T,y^{\text{vapor}}}$ を求めるための変数
Q	p^{residual} (HGK式)
Q5	$\left(\frac{\partial p^{\text{residual}}}{\partial \rho}\right)_T$ (HGK式) であるが、ここでは計算していない。
Q10 (line 16800)	$\rho^2 e^{-\rho}$ (HGK式)
Q10 (line 18500)	$\delta_i^l \exp(-\alpha_i \delta_i^{k_i} - \beta_i \tau_i^2)$ (HGK式)
Q20	$1 - e^{-\rho}$ (HGK式)
QK	k_i (HGK式)
QKM	l_i (HGK式)
QM	$l_i \delta_i^{-1} - \alpha_i k_i \delta_i^{k_i-1}$ (HGK式)
QP(line 17500)	$g_i \rho^2 e^{-\rho} (1 - e^{-\rho})^{k_i-1} \left(\frac{T_0}{T}\right)^{l_i}$ (HGK式)
QP(line 18650)	$\sum_{i=37}^{40} \left[\frac{g_i}{\rho_i} (l_i \delta_i^{-1} - \alpha_i k_i \delta_i^{k_i-1}) \delta_i^l \rho^2 \exp(-\alpha_i \delta_i^{k_i} - \beta_i \tau_i^2) \right]$ (HGK式)
QR(I)	$\rho^2 e^{-\rho} (1 - e^{-\rho})^{i-2}$ (HGK式)
QT(I)	$\left(\frac{T_0}{T}\right)^{i-2}$ (HGK式)
QZR(I)	$\rho^2 e^{-\rho} (1 - e^{-\rho})^i$ (HGK式)
QZT(I)	$\left(\frac{T_0}{T}\right)^{i-1}$ (HGK式)
R	気体定数(= 8.3144 J K ⁻¹ mol ⁻¹)
RHOC	臨界点における純水の密度(= 0.322 g cm ⁻³)
RHOL	液相の密度の計算値(g cm ⁻³)
RHOLD	$\left(\frac{\partial p^{\text{liquid}}}{\partial r^{\text{liquid}}}\right)_{T,y^{\text{liquid}}}$ を求めるための変数
RHOLIQUID	液相の密度(g cm ⁻³)

RHOV	気相の密度の計算値(g cm ⁻³)
RHOVAPOR	気相の密度(g cm ⁻³)
RHOVD	$\left(\frac{\partial p^{\text{vapor}}}{\partial r^{\text{vapor}}}\right)_{T,y^{\text{vapor}}}$ を求めるための変数
RT	RT (J mol ⁻¹)
RTG	RT (J g ⁻¹) (HGK式)
S	気相と液相と固相が平衡状態にある時の液相中の塩化ナトリウムの濃度 (wt.%)
SREF	HGK式では基準状態を三重点での液相にしている。この時にエントロピーの計算値が0になるようにしている。Haar et al. (1984)はサブルーチン THERMDTで無次元化したエントロピー(SD)の計算値からSREFの値を引いている。三重点でのエントロピーの計算値が0になるようにするために Haar et al. (1984)はSREFの値を7.6180802とおいた。本計算プログラムでは、SREFの値を7.6180720にして基準状態の計算値を0に近い値にした。
T	絶対温度で表示した入力温度
TAU(line 12100)	Wagner and Pruss (1993)の式で用いる温度に関する変数
TAU(line 18350)	τ_i (HGK式)
TC	純水の臨界温度(= 647.096 K)
TEX	$\exp(-\beta_i \tau_i^2)$ (HGK式)
TIDEAL	$T/100$ (HGK式)
TL	$\log(T/100)$ (HGK式)
TT	温度(°C)の入力値
TTT	絶対温度で表示した入力温度 (HGK式)
TX	T_i (HGK式)
TZ	647.073 (HGK式)
UREF	HGK式では基準状態を三重点での液相にしている。この時に内部エネルギーの計算値が0になるようにしている。Haar et al. (1984)はサブルーチン THERMDTで無次元化した内部エネルギー(UD)の計算値からUREFの値を絶対温度で割った値を引いている。三重点での内部エネルギーの計算値が0になるようにするために Haar et al. (1984)はUREFの値を-4328.455039とおいた。本計算プログラムでは、UREFの値を-4328.454977にして基準状態の計算値を0に近い値にした。
V	$\frac{647.073}{T}$ (HGK式)
V0	純水が気液二相平衡条件下にある時の液相の比体積(cm ³ g ⁻¹)
V(1),V(2),V(3),V(4), V(5),V(6),V(7),V(8), V(9),V(10)	b や \bar{B} を求めるための係数 (HGK式)

VC	純水の臨界点におけるモル体積の値をJ bar mol ⁻¹ の単位で表した値。単位をcm ³ mol ⁻¹ に取っていない関係でプログラムのline 10450では変数H2Oを変数RHOCで割った値をさらに10で割って変数VCの値としている。
VCON	Haas式における定数(= 3.106 cm ³ g ⁻¹)
WL1,WL2,WL3,WL4, WL5,WL6	純水が気液二相平衡条件下にある時に液相の密度をWagner-Pruss式で計算するための係数
WLNACL	液相中の塩化ナトリウムの濃度(wt%)
WV1,WV2,WV3,WV4, WV5,WV6	純水が気液二相平衡条件下にある時に気相の密度をWagner-Pruss式で計算するための係数
X	純水の密度 (HGK式)
XL	液相中での塩化ナトリウムのモル分率
XV	気相中での塩化ナトリウムのモル分率
XX	1 - y (HGK式)
Y	bp/4 (HGK式)
YL	液相中でH ₂ O 1 モルに溶解している塩化ナトリウムの物質質量 (モル)
YV	気相中でH ₂ O 1 モルに溶解している塩化ナトリウムの物質質量 (モル)
YVDEC	y ^{vapor} の初期推定値を求めるための変数
YVMAX	YVの値の最大値
YVMIN	YVの値の最小値
Z (line 19050)	p _{base} /ρRT (HGK式)
Z (line 19350)	p/ρRT (HGK式)
Z0	$\frac{1 + \alpha y + \beta y^2}{(1 - y)^3}$ (HGK式)

4. Program list

```
10000 REM Calculation of vapor-liquid equilibrium for aqueous NaCl solution
10050 REM Tanger and Pitzer (1989)
10100 DEFDBL A-H, M-Z
10150 DIM HGKG(40), II(40), JJ(40), BP(10), BQ(10)
10200 DIM ATZ(4), ADZ(4), AAT(4), AAD(4)
10250 DIM COEF(10, 10), BBC(10), BC(10)
10300 DIM QR(11), QT(10), QZR(9), QZT(9), V(10), A(8), C(18)
10350 GOSUB *BLOCKDATA
10400 H2O=18.01534:NACL=58.4428#:R=8.3144#
10450 RHOC=.322#:VC=H2O/(RHOC*10)
10500 INPUT"Input temperature (deg C)";TT
10550 IF TT<250 THEN GOTO 10500
10600 T=TT+273.15#
10650 TTT=T
10700 RT=R*T
10750 INPUT "Input molality of NaCl in liquid phase";M
10760 IF M=<0 THEN GOTO 10750
10800 LPRINT USING"Temperature(deg C)=####.###      molality of NaCl=+#.#####";TT,M
10850 LPRINT
10900 WLNACL=100*NACL*M/(1000+NACL*M)
10950 REM Bischoff (1991)
11000 S=23.637#+.057798*TT-.0002132*TT*TT+7.5213D-007*TT*TT*TT-5.355D-010*TT*TT*TT*TT
11050 PBISCHOFF=41.749#-1.2125#*TT+.0136213*TT*TT-7.52333D-005*TT*TT*TT+2.19664D-007*TT*TT*TT*TT
11100 PBISCHOFF=PBISCHOFF-2.82583D-010*TT*TT*TT*TT*TT+1.27231D-013*TT*TT*TT*TT*TT*TT
11150 IF WLNACL<S THEN GOTO 11400
11200 LPRINT "Supersaturated (or saturated) with NaCl"
11250 LPRINT USING"Pressure = +#.###^#### Solubility=+#.##";PBISCHOFF, 1000*S/((100-S)*NACL)
11300 LPRINT : LPRINT : LPRINT : LPRINT
11350 GOTO 10500
11400 XL=M/(M+1000/H2O)
11450 YL=M*H2O/1000
11500 INPUT"Do you input initial estimates for DRL, DRV, and YV? Input Y (or y) or N (n)";A$
11550 IF A$="N" OR A$="n" THEN GOTO 12000
11600 INPUT"DRL";DRL:INPUT"DRV";DRV:INPUT"YV";YV
11650 DELDL=1D-006:DELDV=1D-006
11700 MYULS=0:MYUVS=0:MYULWATER=0:MYUVWATER=0
11750 REM Parameters of Tanger and Pitzer
11800 B10=-29984.4#+19.0285*T+6.65541#*(10000/T)*(10000/T)*(10000/T)-1.20069*1000*(1000/T)*(1000/T)*(1000/T)*
(1000/T)*(1000/T)
11850 B11=3928.3#-10.5947*T-.60751*(1000/T)*(1000000/(T*T))*(1000000/(T*T))*(1000000/(T*T))*(1000000/(T*T))*
(1000000/(T*T))*(1000000/(T*T))
11900 B20=14121.9#-27.0731*T-257.142#*(1000/T)*(1000000/(T*T))*(1000000/(T*T))*(1000000/(T*T))
11950 GOTO 14500
12000 TC=647.096
12050 IF TT=<350 THEN GOTO 12150
12100 TAU=1#-623.15#/TC:GOTO 12200
12150 TAU=1#-T/TC
12200 GOSUB *WPNACL
12250 DRLO=DAQ*(1#-WLNACL/100)/RHOC:DRVO=DWATERV/RHOC
12300 XV=1D-012:YV=XV/(1#+XV)
12350 DRL=0:DRV=0
12400 DELDL=1D-006:DELDV=1D-006
12450 MYULS=0:MYUVS=0:MYULWATER=0:MYUVWATER=0
12500 REM Parameters of Tanger and Pitzer
12550 B10=-29984.4#+19.0285*T+6.65541#*(10000/T)*(10000/T)*(10000/T)-1.20069*1000*(1000/T)*(1000/T)*(1000/T)*
(1000/T)*(1000/T)
12600 B11=3928.3#-10.5947*T-.60751*(1000/T)*(1000000/(T*T))*(1000000/(T*T))*(1000000/(T*T))*(1000000/(T*T))*
```

```

(1000000/(T*T))*(1000000/(T*T))
12650 B20=14121.9#-27.0731*T-257.142#*(1000/T)*(1000000/(T*T))*(1000000/(T*T))*(1000000/(T*T))
12700 DRL=DRLO
12750 DRV=DRVO
12800 MYULS=RT*LOG(XL)+VC*((-1#)*B10/DRL+B11*(LOG(DRL)+1#/DRL))-2#*VC*YL*B20/DRL
12850 MYUVS=RT*LOG(YV)-RT*LOG(1#+YV)+VC*((-1#)*B10/DRV+B11*(LOG(DRV)+1#/DRV))-2#*VC*YV*B20/DRV
12900 DIFMYUS=MYUVS-MYULS
12950 DMYUSDY=RT/YV/(1#+YV)-2#*VC*B20/DRV
13000 IF ABS(DIFMYUS)<1D-010 THEN GOTO 13300
13050 IF ABS(DIFMYUS/DMYUSDY)<1D-021 THEN GOTO 13300
13100 YVDEC=DIFMYUS/DMYUSDY
13150 IF YVDEC<YV THEN GOTO 13250
13200 YVDEC=YVDEC*.5#:GOTO 13150
13250 YV=YV-YVDEC:GOTO 12850
13300 XV=YV/(1#+YV)
13350 X=0: PRES=0: PL=0: PV=0: PLD=0
13400 RHOV=DRV*RHOC
13450 X=RHOV
13500 GOSUB *WATER
13550 PV=PRES
13600 PV=PV+YV*(B10+B11*(DRV-1#))+YV*YV*B20
13650 RHOL=DRL*RHOC
13700 X=RHOL
13750 GOSUB *WATER
13800 PL=PRES
13850 PL=PL+YL*(B10+B11*(DRL-1#))+YL*YL*B20
13900 DIFPLV=PL-PV
13950 RHOLD=(DRL+DELDL)*RHOC
14000 X=RHOLD
14050 GOSUB *WATER
14100 PLD=PRES
14150 PLD=PLD+YL*(B10+B11*(DRL+DELDL)-1#)+YL*YL*B20
14200 DPDDL=(PLD-PL)/DELDL
14250 IF ABS(DIFPLV/DPDDL)<1D-010 THEN GOTO 14500
14300 DRLDEC=DIFPLV/DPDDL
14350 IF DRL>DRLDEC THEN GOTO 14450
14400 DRLDEC=.5#*DRLDEC:GOTO 14350
14450 DRL=DRL-DRLDEC:GOTO 13650
14500 GOSUB *SOLUTION
14550 LPRINT
14600 DRL=BC(1): DRV=BC(2): YV=BC(3)
14650 GOSUB *FUNC
14700 RHOLIQUID=(H20+YL*NACL)*RHOC*DRL/H20
14750 RHOVAPOR=(H20+YV*NACL)*RHOC*DRV/H20
14800 LPRINT USING"Pressure liquid=+#.###^vapor=+#.###^";PL,PV
14850 LPRINT USING"Density liquid=+#.###^vapor=+#.###^";RHOLIQUID,RHOVAPOR
14900 LPRINT :LPRINT :LPRINT:LPRINT
14950 GOTO 10500
15000 *WPNACL
15050 REM Densities of liquid and vapor under vapor-saturation
15100 WL1=1.99274064:WL2=1.09965342:WL3=-.510839303:WL4=-1.75493479#:WL5=-45.5170352:WL6=-674694.45#
15150 WV1=-2.0315024:WV2=-2.6830294:WV3=-5.38626492#:WV4=-17.2991605:WV5=-44.7586581#:WV6=-63.9201063
15200 DL=1#+WL1*EXP((1#/3#)*LOG(TAU))+WL2*EXP((2#/3#)*LOG(TAU))+WL3*EXP((5#/3#)*LOG(TAU))+WL4*EXP((16#/3#)*LOG(TAU))+WL5*EXP((43#/3#)*LOG(TAU))+WL6*EXP((110/3#)*LOG(TAU))
15250 DL=RHOC*DL
15300 DV=WV1*EXP((2#/6#)*LOG(TAU))+WV2*EXP((4#/6#)*LOG(TAU))+WV3*EXP((8#/6#)*LOG(TAU))+WV4*EXP((18#/6#)*LOG(TAU))+WV5*EXP((37#/6#)*LOG(TAU))+WV6*EXP((71#/6#)*LOG(TAU))
15350 DV=RHOC*EXP(DV):DWATERV=DV
15400 BA1=-398.216#:BA2=821.77#:BA3=-407.86:BA4=-44.9495#:BA5=51.3658#

```

```
15450 VCON=3.106
15500 VO=1#/DL
15550 D1=1000*VO+M*BA1+M*BA2*VO+M*BA3*VO*VO+(BA4+BA5*VO)*VO*VO*M*SQR(M)/((VCON-VO)*(VCON-VO))
15600 DAQ=(1000+NACL*M)/D1
15650 RETURN
15700 *WATER
15750 GIB=0
15800 RTG=GASCON*T
15850 REM BBT
15900 V(1)=1#
15950 FOR I=2 TO 10
16000 V(I)=V(I-1)*TZ/T
16050 NEXT I
16100 B1=BP(1)+BP(2)*LOG(1#/V(2))
16150 B2=BQ(1)
16200 FOR I=3 TO 10
16250 B1=B1+BP(I)*V(I-1)
16300 B2=B2+BQ(I)*V(I-1)
16350 NEXT I
16400 DD=X
16450 D=DD*FD
16500 REM QQTD
16550 QR(1)=0
16600 Q5=0
16650 Q=0
16700 AR=0
16750 E=EXP(-AA*D)
16800 Q10=D*D*E
16850 Q20=1#-E
16900 QR(2)=Q10
16950 V=TZ/T
17000 QT(1)=T/TZ
17050 FOR I=2 TO 10
17100 QR(I+1)=QR(I)*Q20
17150 QT(I)=QT(I-1)*V
17200 NEXT I
17250 FOR I=1 TO INC
17300 K=II(I)+1
17350 L=JJ(I)
17400 QK=CDBL(K)
17450 QZR(K-1)=QR(K+1):QZT(L)=QT(L+1):QZR(K)=QR(K+2):QZT(L+1)=QT(L+2)
17500 QP=HGKG(I)*AA*QZR(K-1)*QZT(L)
17550 Q=Q+QP
17600 AR=AR+HGKG(I)*QZR(K)*QZT(L)/(Q10*QK*RTG)
17650 NEXT I
17700 QP=0
17750 FOR J=37 TO 40
17800 IF HGKG(J)=0 THEN GOTO 18900
17850 K=II(J)
17900 KM=JJ(J)
17910 QK=CDBL(K):QKM=CDBL(KM)
17950 DDZ=ADZ(J-36)
18000 DEL=D/DDZ-1#
18050 IF ABS(DEL)<1D-010 THEN DEL=1D-010
18150 EX1=-AAD(J-36)*(DEL^QK)
18200 DEX=EXP(EX1)*(DEL^QKM)
18250 ATT=AAT(J-36)
18300 TX=ATZ(J-36)
18350 TAU=T/TX-1#
```

```
18400 EX2=-ATT*TAU*TAU
18450 TEX=EXP(EX2)
18500 Q10=DEX*TEX
18550 QM=QKM/DEL-QK*AAD(J-36)*DEL^(QK-1#)
18600 FCT=QM*D*D*Q10/DDZ
18650 QP=QP+HGKG(J)*FCT
18700 AR=AR+Q10*HGKG(J)/RTG
18750 NEXT J
18800 Q=Q+QP
18850 REM BASEDT
18900 Y=.25#*B1*D
18950 XX=1#-Y
19000 Z0=(1#+G1*Y+G2*Y*Y)/(XX*XX*XX)
19050 Z=Z0+4#*Y*(B2/B1-GF)
19100 AB=-LOG(XX)-(G2-1#)/XX+28.16666667#/(XX*XX)+4#*Y*(B2/B1-GF)+15.166666667#+LOG(D*RTG/.101325)
19200 BASEF=Z
19300 PRES=FP*(RTG*D*BASEF+Q)
19350 Z=BASEF+Q/(RTG*D)
19400 REM THERMDT
19450 TIDEAL=T/100
19500 TL=LOG(TIDEAL)
19550 GI=- (C(1)/TIDEAL+C(2))*TL
19600 FOR I=3 TO 18
19650 GI=GI-C(I)*(TIDEAL^CDBL((I-6)))
19700 NEXT I
19750 AI=GI-1#
19800 AD=AB+AR+AI-UREF/T+SREF
19850 GD=AD+Z
19900 GIB=GD*RTG*FH
19950 RETURN
20000 *FUNC
20050 FSUMI=0
20100 GIB=0: PRES=0: PV=0: PL=0: X=0: DIFP=0: DIFMYUS=0: DIFMYUWATER=0
20150 DRL=BBC(1): DRV=BBC(2): YV=BBC(3)
20200 RHOV=DRV*RHOC
20250 X=RHOV
20300 GOSUB *WATER
20350 GIBV=GIB
20400 PV=PRES
20450 PV=PV+YV*(B10+B11*(DRV-1#))+YV*YV*B20
20500 RHOL=DRL*RHOC
20550 X=RHOL
20600 GOSUB *WATER
20650 GIBL=GIB
20700 PL=PRES
20750 PL=PL+YL*(B10+B11*(DRL-1#))+YL*YL*B20
20800 DIFP=PV-PL
20850 RHOVD=(DRV+DEL DV)*RHOC
20900 X=RHOVD
20950 GOSUB *WATER
21000 GIBVD=GIB
21050 PVD=PRES
21100 PVD=PVD+YV*(B10+B11*((DRV+DEL DV)-1#))+YV*YV*B20
21150 DPDDV=(PVD-PV)/DEL DV
21200 DPDY=B10+B11*(DRV-1#)+2#*YV*B20
21250 RHOLD=(DRL+DEL DL)*RHOC
21300 X=RHOLD
21350 GOSUB *WATER
21400 GIBLD=GIB
```

```
21450 PLD=PRES
21500 PLD=PLD+YL*(B10+B11*((DRL+DELDL)-1#))+YL*YL*B20
21550 DPDDL=(PLD-PL)/DELDL
21600 MYULS=RT*LOG(XL)+VC*((-1#)*B10/DRL+B11*(LOG(DRL)+1#/DRL))-2#*VC*YL*B20/DRL
21650 MYUVS=RT*LOG(YV)-RT*LOG(1#+YV)+VC*((-1#)*B10/DRV+B11*(LOG(DRV)+1#/DRV))-2#*VC*YV*B20/DRV
21700 DIFMYUS=MYUVS-MYULS
21750 DMYUSDDL=VC*B11/DRL+VC*(B10-B11+2#*YL*B20)/(DRL*DRL)
21800 DMYUSDDV=VC*B11/DRV+VC*(B10-B11+2#*YV*B20)/(DRV*DRV)
21850 DMYUSDY=RT/YV/(1#+YV)-2#*VC*B20/DRV
21900 MYULWATER=GIBL+YL*VC*(B10/DRL+B11*(1#-1/DRL))+2#*VC*YL*YL*B20/DRL+RT*LOG(1#-XL)
21950 MYUVWATER=GIBV+YV*VC*(B10/DRV+B11*(1#-1/DRV))+2#*VC*YV*YV*B20/DRV-RT*LOG(1#+YV)
22000 DIFMYUWATER=MYUVWATER-MYULWATER
22050 DMYUWATERDDL=(GIBLD-GIBL)/DELDL-YL*VC*(B10-B11+2#*YL*B20)/(DRL*DRL)
22100 DMYUWATERDDV=(GIBVD-GIBV)/DELDV-YV*VC*(B10-B11+2#*YV*B20)/(DRV*DRV)
22150 DMYUWATERDY=VC*(B10/DRV+B11*(1#-1/DRV))+4#*YV*VC*B20/DRV-RT/(1#+YV)
22200 FSUMI=(PV/PL-1#)*(PV/PL-1#)+(MYUVS/MYULS-1#)*(MYUVS/MYULS-1#)+(MYUVWATER/MYULWATER-1#)*(MYUVWATER/MYULWATER-1#)
22250 RETURN
22300 *SOLUTION
22350 REM Non-linear equation
22400 REM Main routine
22450 REM Unknowns=KK
22500 EPS=1D-005:IMAX=100:AL=1#:KK=3
22550 DRLMAX=3.5#:DRLMIN=1.5#:DRVMAX=1#:DRVMIN=.02:YVMIN=1D-012
22600 YVMAX=YL-.001
22650 IF TT<350 THEN GOTO 22850
22700 IF TT>=540 THEN GOTO 22800
22750 DRLMIN=.5#:DRVMAX=2#:GOTO 22850
22800 DRLMIN=.5#:DRVMAX=2.5#
22850 LPRINT USING"Eps=+#.##^";EPS
22900 LPRINT USING"DRLMAX=+#.## DRLMIN=+#.## DRVMAX=+#.## DRVMIN=+#.## YVMAX=+#.##^ YVMIN=+#.##^";DRLMAX,DRLMIN,DRVMAX,DRVMIN,YVMAX,YVMIN
22950 BC(1)=DRL:BC(2)=DRV:BC(3)=YV
23000 LPRINT"Initial estimate"
23050 FOR I=1 TO KK
23100 LPRINT USING"BC(#)=+#.###^";I,BC(I);
23150 NEXT I
23200 LPRINT
23250 FOR I=1 TO KK:BBC(I)=BC(I):NEXT I
23300 GOSUB *FUNC
23350 LPRINT USING"Initial FSUM=+#.#####^";FSUMI
23400 LPRINT
23450 FSUM=FSUMI
23500 ICON=KK:ITER=1
23550 GOSUB *BCSOLV
23600 LPRINT USING"ICON=## FSUM=+#.#####^ ITERATION=###";ICON,FSUM,ITER
23650 ITER=ITER+1:AL=1#
23700 FOR J=1 TO KK:LPRINT USING"BC(#)=+#.#####^";J,BC(J);:NEXT J:LPRINT
23750 IF ITER>IMAX THEN ICON=-1
23800 LPRINT
23850 IF ICON>0 THEN GOTO 23550
23900 IF ICON=0 THEN GOTO 24050
23950 IF ICON=-1 THEN GOTO 24000
24000 LPRINT "Function improvement possible"
24050 LPRINT "Solution"
24100 FOR J=1 TO KK
24150 LPRINT USING"BC(#)=+#.#####^";J,BC(J)
24200 NEXT J
24250 RETURN
```



```
24300 REM Subroutine BCSOLV
24350 *BCSOLV
24400 KP1=KK+1
24450 FOR I=1 TO KK:FOR J=1 TO KP1:COEF(I, J)=0:NEXT J:NEXT I
24500 FOR I=1 TO KK:BBC(I)=BC(I):NEXT I
24550 GOSUB *FUNC
24600 COEF(1, KP1)=(-1#)*DIFP:COEF(2, KP1)=(-1#)*DIFMYUS:COEF(3, KP1)=(-1#)*DIFMYUWATER
24650 COEF(1, 1)=(-1#)*DPDDL:COEF(1, 2)=DPDDV:COEF(1, 3)=DPDY
24700 COEF(2, 1)=(-1#)*DMYUSDDL:COEF(2, 2)=DMYUSDDV:COEF(2, 3)=DMYUSDY
24750 COEF(3, 1)=(-1#)*DMYUWATERDDL:COEF(3, 2)=DMYUWATERDDV:COEF(3, 3)=DMYUWATERDY
24800 REM Solve correction equation
24850 FOR ISL1=1 TO KK
24900 ISL2=ISL1+1
24950 FOR ISL3=ISL2 TO KP1
25000 COEF(ISL1, ISL3)=COEF(ISL1, ISL3)/COEF(ISL1, ISL1)
25050 NEXT ISL3
25100 FOR ISL3=1 TO KK
25150 IF ISL1<>ISL3 THEN GOTO 25250
25200 GOTO 25400
25250 FOR ISL4=ISL2 TO KP1
25300 COEF(ISL3, ISL4)=COEF(ISL3, ISL4)-COEF(ISL1, ISL4)*COEF(ISL3, ISL1)
25350 NEXT ISL4
25400 NEXT ISL3
25450 NEXT ISL1
25500 FOR J1=1 TO KK
25550 BBC(J1)=BC(J1)+COEF(J1, KP1)
25600 NEXT J1
25650 ICON=0
25700 IF BBC(1)>DRLMIN THEN GOTO 25800
25750 BBC(1)=DRLMIN:GOTO 25900
25800 IF BBC(1)<DRLMAX THEN GOTO 25900
25850 BBC(1)=DRLMAX
25900 IF BBC(2)>DRVMIN THEN GOTO 26000
25950 BBC(2)=DRVMIN:GOTO 26100
26000 IF BBC(2)<DRVMAX THEN GOTO 26100
26050 BBC(2)=DRVMAX
26100 IF BBC(3)>YVMIN THEN GOTO 26200
26150 BBC(3)=YVMIN:GOTO 26300
26200 IF BBC(3)<YVMAX THEN GOTO 26300
26250 BBC(3)=YVMAX
26300 GOSUB *FUNC
26350 IF FSUM<FSUM THEN FSUM=FSUM:GOTO 27500
26400 AL=AL*1#/2#
26450 FOR J1=1 TO KK
26500 BBC(J1)=BC(J1)+AL*COEF(J1, KP1)
26550 NEXT J1
26600 IF BBC(1)>DRLMIN THEN GOTO 26700
26650 BBC(1)=DRLMIN:GOTO 26800
26700 IF BBC(1)<DRLMAX THEN GOTO 26800
26750 BBC(1)=DRLMAX
26800 IF BBC(2)>DRVMIN THEN GOTO 26900
26850 BBC(2)=DRVMIN:GOTO 27000
26900 IF BBC(2)<DRVMAX THEN GOTO 27000
26950 BBC(2)=DRVMAX
27000 IF BBC(3)>YVMIN THEN GOTO 27100
27050 BBC(3)=YVMIN:GOTO 27200
27100 IF BBC(3)<YVMAX THEN GOTO 27200
27150 BBC(3)=YVMAX
27200 GOSUB *FUNC
```

```
27250 IF FSUMI<FSUM THEN FSUM=FSUMI:GOTO 27500
27300 AL=AL*.5#
27350 IF AL>1D-006 OR AL<-1D-006 THEN GOTO 26450
27400 IF AL>=-1D-006 AND AL<0 THEN LPRINT "Failed.":END
27450 AL=-1:GOTO 26450
27500 FOR I=1 TO KK
27550 IF ABS(COEF(I,KP1)/BBC(I))>=EPS THEN ICON=ICON+1
27600 NEXT I
27650 FOR I=1 TO KK:BC(I)=BBC(I):NEXT I
27700 RETURN
27750 *BLOCKDATA
27800 FOR I=1 TO 4:READ ATZ(I):NEXT I
27850 DATA 640#,640#,641.6#,270#
27900 FOR I=1 TO 4:READ ADZ(I):NEXT I
27950 DATA 0.319#,0.319#,0.319#,1.55#
28000 FOR I=1 TO 4:READ AAT(I):NEXT I
28050 DATA 2.0D+004,2.0D+004,4.0D+004,25.0#
28100 FOR I=1 TO 4:READ AAD(I):NEXT I
28150 DATA 34.0#,40.0#,30.0#,1050#
28200 GASCON=.461522#:TZ=647.073:AA=1#:INC=36
28250 UREF=-4328.454977#:SREF=7.6180720#
28300 G1=11#:G2=44.33333333333333#:GF=3.5#
28350 FOR I=1 TO 10:READ BP(I):NEXT I
28400 DATA 0.7478629#,-0.3540782#,0.0#,0.0#,0.007159876#,0.0#,-0.003528426#,0.0#,0.0#,0.0#
28450 FOR I=1 TO 10:READ BQ(I):NEXT I
28500 DATA 1.1278334#,0.0#,-0.5944001#,-5.010996#,0.0#,0.63684256#,0.0#,0.0#,0.0#,0.0#
28550 FOR I=1 TO 40:READ HGKG(I):NEXT I
28600 DATA -5.3062968529023D+002,2.2744901424408D+003,7.8779333020687D+002
28650 DATA -6.9830527374994D+001,1.7863832875422D+004,-3.9514731563338D+004
28700 DATA 3.3803884280753D+004,-1.3855050202703D+004,-2.5637436613260D+005
28750 DATA 4.8212575981415D+005,-3.4183016969660D+005,1.2223156417448D+005
28800 DATA 1.1797433655832D+006,-2.1734810110373D+006,1.0829952168620D+006
28850 DATA -2.5441998064049D+005,-3.137774947767D+006,5.2911910757704D+006
28900 DATA -1.3802577177877D+006,-2.5109914369001D+005,4.6561826115608D+006
28950 DATA -7.2752773275387D+006,4.1774246148294D+005,1.4016358244614D+006
29000 DATA -3.1555231392127D+006,4.7929666384584D+006,4.0912664781209D+005
29050 DATA -1.3626369388386D+006,6.9625220862664D+005,-1.0834900096447D+006
29100 DATA -2.2722827401688D+005,3.8365486000660D+005,6.8833257944332D+003
29150 DATA 2.1757245522644D+004,-2.6627944829770D+003,-7.0730418082074D+004
29200 DATA -0.225#,-1.68#,0.055#,-93.0#
29250 FOR I=1 TO 40:READ II(I):NEXT I
29300 DATA 0,0,0,0,1,1,1,1,2,2,2,2,3,3,3,3,4,4,4,4,5,5,5,5,6,6,6,6,8,8,8,8,2,2,0,4,2,2,2,4
29350 FOR I=1 TO 40:READ JJ(I):NEXT I
29400 DATA 2,3,5,7,2,3,5,7,2,3,5,7,2,3,5,7,2,3,5,7,2,3,5,7,2,3,5,7,1,4,4,4,0,2,0,0
29450 FOR I=1 TO 18:READ C(I):NEXT I
29500 DATA 1.9730271018D+001,2.09662681977D+001,-4.83429455355D-001,6.05743189245D+000,2.256023885D+001,-9.87
532442D+000,-4.3135538513D+000,4.58155781D-001
29550 DATA -4.7754901883D-002,4.1238460633D-003,-2.7929052852D-004
29600 DATA 1.4481695261D-005,-5.6473658748D-007,1.6200446D-008,-3.303822796D-010,4.51916067368D-012,-3.707341
22708D-014,1.37546068238D-016
29650 FD=1#:FP=10:FH=18.01534
29700 RETURN
```

文献

澁江靖弘 (2012) 250°Cから600°Cにおける塩化ナトリウム—水系の気液平衡. 兵庫教育大学研究紀要, **41**, 57–68.